
%_mprintto; 

options notes nosource; 

proc datasets lib=work nolist memtype=data kill; quit; 

%put NOTE: 

=======================================================================; 

%put NOTE: Covance Study Number : 000000106326; 

%put NOTE: Client Protocol ID   : ZRHM-PK-05-JP; 

%put NOTE: Program Name         : t_qsu.sas; 

%put NOTE: Purpose              : table of QSU factors and total score; 

%put NOTE: ; 

%put NOTE: Input Data           : ADAM.ADSL ADAM.ADQSSU; 

%put NOTE: Output               : t_15_2_4_14(qsu); 

%put NOTE: Macros Called        : _MPRINTTO; 

%put NOTE: ; 

%put NOTE: Programmed by        : cvn_jriley; 

%put NOTE: Creation Date        : 2014-08-06; 

%put NOTE: SAS Version          : 9.3; 

%put NOTE: ; 

%put NOTE: == Latest Run 

=========================================================; 

%put NOTE: Run by               : &sysuserid; 

%put NOTE: Date/Time            : 

%sysfunc(putn(%sysfunc(date()),e8601da.))T%sysfunc(putn(%sysfunc(time()),

e8601lz.)); 

%put NOTE: ; 

%put NOTE: == Modification History 

===============================================; 

%put NOTE: Date        Initials   No. Reason; 

%put NOTE: 11Aug2014   JR         1)  Amended footnote;  

%put NOTE: 23Sep2014   JR         2)  Added QSU footnote;  

%put NOTE: ; 

%put NOTE: 

=======================================================================; 

options notes source source2 nofullstimer validvarname=upcase missing=' 

'; 

ods _all_ close; 

ods listing; 

 

*=====================================================================; 

* START OF PROGRAM CODE                                               ; 

*=====================================================================; 

 

%let tflno=T_15_02_04_14(qsu);  

 

 

%let TFL_Part=%scan(&_SASPROGRAMFILE,-3,%str(/)); 

 

data _null_; 

     tmp="&TFL_Part"; 

  if tmp not in ("dev" "qc") then call symput("TFL_Part", "prod"); 

  call symput('TFLpath', compress("&_SASPROGRAMFILE","'")); 

run; 

 

******************************; 



* read in data ; 

******************************; 

 

/* Calculate totals for products */ 

data adsl; 

 set adam.adsl(where=(pprotfl='Y')); 

 if analgr1n=1 then do; 

  if index(trt01a,'THS 2.2') or index(trt02a,'THS 2.2') then 

trtord=1; 

  output; 

  if index(trt01a,'CC') or index(trt02a,'CC') then trtord=2; 

  output; 

 end; 

 else if analgr1n=2 then do; 

  if index(trt01a,'THS 2.2') or index(trt02a,'THS 2.2') then 

trtord=10; 

  output; 

  if index(trt01a,'NRT gum') or index(trt02a,'NRT gum') then 

trtord=7; 

  output; 

 end; 

 else if missing(analgr1n) then delete; 

run; 

 

proc sort data=adsl nodupkey out=adsl1; 

 by analgr1n analgr1 trtord subjid; 

run; 

 

proc freq data=adsl1(where=(not missing(trtord))) noprint;   

 table analgr1n*analgr1*trtord/ out =totals2(drop=percent 

rename=(count=total)); 

run; 

 

data totals3; 

    set totals2; 

 

    call symput('trt'||strip(put(trtord,best.)),strip(put(total,best.))); 

run; 

 

proc sort data=totals3; 

    by analgr1n analgr1 trtord; 

run; 

 

/*Bring in appropriate data from  ADQSSU*/ 

data adqssu; 

 set adam.adqssu(where=(anl01fl='Y' and pprotfl='Y'));  

run; 

 

data adqssu_orig;  

 set adqssu; 

 

 statval=aval; 

 

    if atpt='T0 + 15 min' or atpt='T0 + 20 min' then do; 



        atpt='T0 + 15/20 min'; 

        atptn=6; 

    end; 

run; 

 

proc sort data=adqssu_orig; 

    by analgr1n analgr1 trtan trta parcat2n parcat2 paramn param paramcd 

atptn atpt; 

run; 

 

proc means data=adqssu_orig alpha=0.05 noprint; 

    var statval; 

    by analgr1n analgr1 trtan trta parcat2n parcat2 paramn param paramcd 

atptn atpt; 

    output out=results02 n=n1 mean=mean1 std=std1 median=median1 min=min1 

max=max1 q1=q1 q3=q3 lclm=lci1 uclm=uci1; 

run; 

 

data results03; 

    set results02; 

    attrib meansd length=$20. 

        minmax length=$20. 

        n      length=$20. 

        median length=$20. 

           ci     length=$20. 

           quart  length=$20.; 

 

    n = left(compress(put(n1,8.))); 

    if not missing(median1) then median  = 

left(compress(put(median1,8.2)));  

 if not missing(mean1) and not missing(std1) then meansd  = 

left(compress(put(round(mean1,0.01),8.2))) || ' (' || 

left(compress(put(0.001*ceil(std1/0.001),8.3))) || ')';  

    if not missing(min1) and not missing(max1) then minmax  = 

left(compress(put(min1,8.1))) || ', ' || left(compress(put(max1,8.1))); 

    if not missing(lci1) and not missing(uci1) then ci = 

strip(strip(compress(put(0.01*floor(lci1/0.01),8.2)))) || ', ' || 

strip(compress(put(0.01*ceil(uci1/0.01),8.2)));  

    if not missing(q1) and not missing(q3) then quart = 

strip(strip(put(q1,8.2)) || ', ' || strip(put(q3,8.2))); 

 

    drop n1 mean1 std1 median1 min1 max1 lci1 uci1 q1 q3; 

run; 

 

data results04; /*Create text as required in output*/ 

    set results03; 

    attrib paramc length = $100.; 

 

    if index(parcat2,'Factor')=0 then paramc=strip(param); 

    else paramc=strip(parcat2); 

 

    if analgr1n=2 and trtan=4 then trtan=10; 

 

    drop param paramcd parcat2 parcat2n; 



run; 

 

proc sort data=results04; 

     by paramn paramc atptn atpt; 

run; 

 

proc transpose data=results04 out=results05 prefix=_ name=varname; 

    by paramn paramc atptn atpt; 

    var n meansd median minmax ci quart; 

    id trtan; 

    idlabel trta; 

run; 

 

data results06; 

    set results05; 

    attrib stat length = $100.; 

 

    if varname='N' then do;  

        statord=1;  

        stat='n';  

    end; 

    else if varname='MEANSD' then do;  

        statord=2;  

        stat='Mean (SD)';  

    end; 

    else if varname='CI' then do;  

        statord=3;  

        stat='95% CI';  

    end; 

    else if varname='MEDIAN' then do;  

        statord=4;  

        stat='Median';  

    end; 

    else if varname='QUART' then do; 

        statord=5; 

        stat='Q25, Q75'; 

    end; 

    else if varname='MINMAX' then do;  

        statord=6;  

        stat='Min, Max';  

    end; 

 

    drop varname; 

run; 

 

data results07; 

    set results06; 

 

    if stat='n' then do; 

        if missing(_4) then _4='0'; 

        if missing(_5) then _2='0'; 

        if missing(_10) then _10='0'; 

        if missing(_7) then _7='0'; 

    end; 



run; 

 

data labels; 

set results07; 

    attrib _4 label = "THS 2.2 Menthol$(N=&trt1)" 

           _5 label = "mCC$(N=&trt2)" 

           _10 label = "THS 2.2 Menthol$(N=&trt10)" 

           _7 label = "NRT gum$(N=&trt7)" 

    ATPT label= "Unformatted timepoint";   

 

 

 if not index(paramc,'-') then 

paramc=upcase(substr(paramc,1,1))||lowcase(substr(paramc,2)); 

 if index(atpt,'T0') then atpt=tranwrd(atpt,'T0','T${sub 0}'); 

run; 

 

proc sort data=labels; 

  by paramn atptn statord; 

run; 

 

/*options replace;*/ 

proc sql noprint; 

 

create table table.t_15_02_04_15 as 

select paramc, atpt, stat, _4, _5, _10, _7 

from labels 

order by paramn, atptn, statord; 

 

quit; 

/*options noreplace;*/ 

 

 

data paging; 

    set labels; 

    by paramn atptn statord; 

 

    flag=1;  

 

    if ln gt 11 then ln=1; /*Amend to look presentable, and avoid page 

overflows*/ 

    else ln+1; 

    if ln=1 then page+1; 

    call symput("page",compress(put(page,best.))); 

run; 

 

options number nodate orientation=landscape papersize=&p_pgsize missing=' 

'; 

ods escapechar='$'; 

%let linetop = \brdrt\brdrs\brdrw30;  * needs to be 1.5pt so calculated 

in twips (1/20 pt) ; 

%let linebot = \brdrb\brdrs\brdrw30; 

%macro outrtf(blankn=, halfblnk=); 

 

%if &halfblnk=N %then %let halfblnk=; 



%else %if &halfblnk=Y %then %let halfblnk=\~; 

 

 

ods path stdlib.t106326 (read) ; 

ods results off; 

ods rtf toc_data 

file="/cvn/projects/prj/data/000000106326/TFL/&TFL_Part./&tflno..rtf" 

style=t106326 startpage=yes headery=1440 footery=1440 ; 

ods noproctitle; 

%do i=1 %to &page; 

 

 

title ; 

footnote; 

%let wd=0; 

ods proclabel=' '; 

 

 

data comp; 

     set paging end=eof;   

  where page=&i; 

      

    /* Amend title as needed */ 

    _firtitl="Table 15.2.4.14 Descriptive Statistics of QSU-brief 

Questionnaire Factors and Total Score - PK Population";    

    _upcas=(length("Path: &TFLpath.")-

length(compress("Path:&TFLpath.",'ABCDEFGHIJKLMNOPQRSTUVWXYZ')))/2; 

    len=&blankn.-length("(page &i of &page)");  

    if eof then do; 

    call symput('_FSRTITL', trim(left(_firtitl))); 

    call symput('_blankn', compress(put(len,best.))); 

    end; 

    drop _firtitl  _upcas len; 

run; 

 

 

ods listing close; 

 

* most set up in template others below; 

* title arial 12pt bold with 12pt paragraph space below; 

* all headers to be arial 11pt bold; 

* data arial 10pt; 

* headers to be central, text values left aligned and numeric centered 

around decimal point; 

proc report data = comp missing headline headskip missing nowd split = 

'$' %if &i=1 %then %do; contents=' ' %end; %else %do; contents='' %end;; 

; 

 column flag page paramn paramc atptn atpt statord stat ("Group-1 PK 

&LINEBOT" _4 _5) ("Group-2 PK &LINEBOT" _10 _7);  

         

    define flag         / order order=internal noprint;  

    define page         / order order = internal noprint; 

    define paramn       / order order = internal noprint; 



    define paramc       / group style={just=left cellwidth=2cm} 

style(header)={just=center} "Variable"; 

    define atptn        / order order=internal noprint; 

    define atpt         / group style={just=left cellwidth=1.5cm} 

style(header)={just=center} "Timepoint"; 

    define statord      / order order = internal noprint; 

    define stat         / display style={just=left cellwidth=1.5cm} 

style(header)={just=center} "Statistic"; 

    define _4           / display style={just=c cellwidth=1.5cm} 

style(header)={just=center};  

    define _5           / display style={just=c cellwidth=1.5cm} 

style(header)={just=center}; 

    define _10          / display style={just=c cellwidth=1.5cm} 

style(header)={just=center}; 

    define _7           / display style={just=c cellwidth=1.5cm} 

style(header)={just=center}; 

 

    break before flag / page %if &i=1 %then %do;  

    contents="&_fsrtitl" %end; %else %do; contents='' %end;;  

 

 break after page / page; 

 

 compute after atptn; 

  line " "; 

 endcomp; 

 

 compute before page / style={protectspecialchars=off};; 

     line "&linetop"; 

  endcomp; 

 

  compute before _page_ / style={just=left protectspecialchars=off};  

     line "\b\fs24\sa24&_FSRTITL." ;  * \b = bold, \fs24 is font 

size 12pt, \sa24 is space after 12pt; 

  line "&linebot"; 

  endcomp; 

 

  compute after _page_/ style={just=left protectspecialchars=off 

PRETEXT="&LINETOP."}; 

   line 'Note: mCC = menthol conventional cigarettes; NRT gum = 

Nicotine Replacement Therapy gum; THS = Tobacco Heating System.';  

   line "Note: T${sub 0} = Time of first product use at single 

use day."; /* 1) JR 11Aug2014 */  

/*   line "Note: T${sub 0} = Time of first product use at single 

use day";*/ 

  line "Note: The QSU-brief values at T${sub 0} + 15/20 min 

refer to QSU-brief data at time T${sub 0} + 15 min for THS 2.2 Menthol 

and mCC, and to QSU-brief data at time T${sub 0} + 20 min for NRT gum."; 

  LINE 'Note: QSU-brief scores reported on a 7-point scale. 

Higher scores indicate greater intensity of urge.'; /* 2) JR 23Sep2014 */  

  line ' '; 

  line 'Appendix 15.3.6.11'; 

  line "Path: &TFLpath." &_blankn.*"\~\~" "(Page &i of &page)"; 

  line "Program Run: &sysdate  &sysuserid   Program Status: 

&status"; 



  endcomp; 

run; 

%end; 

ods rtf close; 

ods results on; 

ods path sashelp.tmplmst (read); 

 

%mend ; 

 

%outrtf(blankn=70, halfblnk=N); 

 

ods listing; 

proc printto print = "&table./t_15_02_04_15.lst" new; 

run; 

 

proc contents data = table.t_15_02_04_15 varnum; 

run; 

ods listing close; 

 

proc printto ; run; 

*=====================================================================; 

* END OF PROGRAM CODE                                                 ; 

*=====================================================================; 

 

 

 

 


